

# Strictly Monotone Brouwer Trees for Well Founded Recursion Over Multiple Values

Anonymous Author(s)

## Abstract

Ordinals can be used to prove the termination of dependently typed programs. Brouwer trees are a particular ordinal notation that make it very easy to assign sizes to higher order data structures. They extend unary natural numbers with a limit constructor, so a function's size can be the least upper bound of the sizes of values from its image. These can then be used to define well founded recursion: any recursive calls are allowed so long as they are on values whose sizes are strictly smaller than the current size.

Unfortunately, Brouwer trees are not algebraically well behaved. They can be characterized equationally as a join-semilattice, where the join takes the maximum of two trees. However, this join does not interact well with the successor constructor, so it does not interact properly with the strict ordering used in well founded recursion.

We present Strictly Monotone Brouwer trees (SMB-trees), a refinement of Brouwer trees that are algebraically well behaved. SMB-trees are built using functions with the same signatures as Brouwer tree constructors, and they satisfy all Brouwer tree inequalities. However, their join operator distributes over the successor, making them suited for well founded recursion or equational reasoning.

This paper teaches how, using dependent pairs and careful definitions, an ill behaved definition can be turned into a well behaved one. Our approach is axiomatically lightweight: it does not rely on Axiom K, univalence, quotient types, or Higher Inductive Types. We implement a recursively-defined maximum operator for Brouwer trees that matches on successors and handles them specifically. Then, we define SMB-trees as the subset of Brouwer trees for which the recursive maximum computes a least upper bound. Finally, we show that every Brouwer tree can be transformed into a corresponding SMB-tree by joining it with itself an infinite number of times. All definitions and theorems are implemented in Agda.

**Keywords:** dependent types, Brouwer trees, well founded recursion

## 1 Introduction

### 1.1 Recursion and Dependent Types

Dependently typed languages, such as Agda [?], Coq [Bertot and Castéran 2004], Idris [?] and Lean [?], bridge the gap between theorem proving and programming.

Functions defined in dependently typed languages are typically required to be *total*: they must provably halt in all inputs. Since the halting problem is undecidable, recursively-defined functions must be written in such a way that the type checker can mechanically deduce termination. Some functions only make recursive calls to structurally-smaller arguments, so their termination is apparent to the compiler. However, some functions cannot be easily expressed using structural recursion. For such functions, the programmer must instead use *well founded recursion*, showing that there is some ordering, with no infinitely-descending chains, for which each recursive call is strictly smaller according to this ordering. For example, the typical quicksort algorithm is not structurally recursive, but can use well founded recursion on the length of the lists being sorted.

### 1.2 Ordinals

While numeric orderings work for first-order data, they are ill suited to recursion over higher-order data structures, where some fields contain functions.

There are many formulations of ordinals in dependent type theory, each with their own advantages and disadvantages.

### 1.3 Contributions

This work defines *strictly monotone Brouwer Trees*, henceforth SMB-trees, a new presentation of ordinals that hit a sort of sweet-spot for defining functions by well founded recursion. Specifically, SMB-trees:

- are strictly ordered by a well founded relation;
- have a maximum operator which computes a least-upper bound;
- are *strictly-monotone* with respect to the maximum: if  $a < b$  and  $c < d$ , then  $\max a c < \max b d$ ;
- can compute the limits of arbitrary sequences;
- are light in axiomatic requirements: they are defined without using axiom K, univalence, quotient types, or higher inductive types.

### 1.4 Uses for SMB-trees

**1.4.1 Well Founded Recursion.** Having a maximum operator for ordinals is particularly useful when traversing over multiple higher order data structures in parallel, where neither argument takes priority over the other. In such a case, a lexicographic ordering cannot be used.

As an example, consider a unification algorithm over some encoding of types, and suppose that  $\alpha$ -renaming or some

other restriction prevents structural recursion from being used. To solve a unification problem  $\Sigma(x : A). B = \Sigma(x : C). D$  we must recursively solve  $A = C$  and  $\forall x. B[x] = D[x]$ . However, the type of  $x$  in the latter equation depends on the solution to the first equation, which is bounded by the size of the maximum of the sizes of both  $A$  and  $C$ . So for each recursive call to be on a smaller size, the size of  $a = c$  and  $b = d$  must both be strictly smaller than  $(a, b) = (c, d)$ . In a lexicographic ordering where the size of the left-hand size dominates, we know that  $a$  is strictly smaller than  $(a, b)$ , but we have no guarantees that  $TODO$ . Conversely, if we order unification problems by the size of the maximum of their two sides.

This style of well founded induction was used to prove termination in a syntactic model of gradual dependent types [?]. There, Brouwer trees were used to establish termination of recursive procedures for combining the type information in two imprecise types. The decreasing metric was the maximum size of the codes for the types being combined. Brouwer trees' arbitrary limits were used to assign sizes to dependent function and product types, and the strict monotonicity of the maximum operator was essential for proving that recursive calls were on strictly smaller arguments.

**1.4.2 Syntactic Models and Sized Types.** An alternate way view of our contribution is as a tool for modelling sized types [?]. The implementation of sized types in Agda has been shown to be unsound [?], due to the interaction between propositional equality and the top size  $\infty$  satisfying  $\infty < \infty$ . [Chan 2022] defines a dependently typed language with sized types that does not have a top size, proving it consistent using a syntactic model based on Brouwer trees.

SMB-trees provide the capability to extend existing syntactic models to sized types with a maximum operator. This brings the capability of consistent sized types closer to feature parity with Agda, which has a maximum operator for its sizes [?], while still maintaining logical consistency.

**1.4.3 Algebraic Reasoning.** Another advantage of SMB-trees is that they allow Brouwer trees to be interpreted using algebraic tools. SMB-trees can be described as In algebraic terminology, SMB-trees satisfy the following algebraic laws, up to the equivalence relation defined by  $s \approx t := s \leq t \leq s$

- Join-semilattice: the binary `max` is associative, commutative, and idempotent
- Bounded: there is a least tree  $Z$  such that `max`  $t Z \approx t$
- Inflationary endomorphism: there is a successor operator  $\uparrow$  such that `max`  $(\uparrow t) \approx \uparrow t$  and  $\uparrow(\text{max } s t) = \text{max}(\uparrow s) (\uparrow t)$

Bezem and Coquand [2022] describe a polynomial time algorithm for solving equations in such an algebra, and describe its usefulness for solving constraints involving universe levels in dependent type checking. While equations involving limits of infinite sequences are undecidable, the

inflationary laws could be used to automatically discharge some equations involving sizes. This algebraic presentation is particularly amenable to solving equations using free extensions of algebras [Allais et al. 2023; Corbyn 2021].

## 1.5 Implementation

We have implemented SMB-trees in Agda 2.6.4. Our library specifically avoids Agda-specific features such as cubical type theory or Axiom K, so we expect that the library can be easily ported to other proof assistants.

This paper is written as a literate Agda document, and the definitions given in the paper are valid Agda code. Several definitions are presented with their body omitted due to space restrictions. The full implementation can be found in the supplementary materials section of this submission.

## 2 Brouwer Trees: An Introduction

Brouwer trees are a simple but elegant tool for proving termination of higher-order procedures. Traditionally, they are defined as follows:

```
data SmallTree : Set where
  Z : SmallTree
  ↑ : SmallTree → SmallTree
  Lim : (ℕ → SmallTree) → SmallTree
```

Under this definition, a Brouwer tree is either zero, the successor of another Brouwer tree, or the limit of a countable sequence of Brouwer trees. However, these are quite weak, in that they can only take the limit of countable sequences. To represent the limits of uncountable sequences, we can parameterize our definition over some Universe à la Tarski:

```
module RawTree {ℓ}
  (C : Set ℓ)
  (El : C → Set ℓ)
  (CN : C) (CNIso : Iso (El CN) ℕ) where
```

Our module is parameterized over a universe level, a type  $\mathbb{C}$  of codes, and an “elements-of” interpretation function  $El$ , which computes the type represented by each code. We require that there be a code whose interpretation is isomorphic to the natural numbers, as this is essential to our construction in ???. Increasingly larger trees can be obtained by setting  $\mathbb{C} := \text{Set } \ell$  and  $El := id$  for increasing  $\ell$ . However, by defining an inductive-recursive universe, one can still capture limits over some non-countable types, since `Tree` is in `Set` whenever  $\mathbb{C}$  is.

We then generalize limits to any function whose domain is the interpretation of some code.

```
data Tree : Set ℓ where
  Z : Tree
  ↑ : Tree → Tree
  Lim : ∀ (c : C) → (f : El c → Tree) → Tree
```

The small limit constructor can be recovered from the natural-number code

```
INLim : (ℕ → Tree) → Tree
INLim f = Lim CN (λ cn → f (Iso.fun CNIso cn))
```

Brouwer trees are a the quintessential example of a higher-order inductive type.<sup>1</sup> Each tree is built using smaller trees or functions producing smaller trees, which is essentially a way of storing a possibly infinite number of smaller trees.

## 2.1 Ordering Trees

Our ultimate goal is to have a well-founded ordering<sup>2</sup>, so we define a relation to order Brouwer trees.

```
data _≤_ : Tree → Tree → Set ℓ where
  ≤-Z : ∀ {t} → Z ≤ t
  ≤-sucMono : ∀ {t1 t2}
    → t1 ≤ t2
    → ↑ t1 ≤ ↑ t2
  ≤-cocone : ∀ {t} {c : C} (f : El c → Tree) (k : El c)
    → t ≤ f k
    → t ≤ Lim c f
  ≤-limiting : ∀ {t} {c : C}
    → (f : El c → Tree)
    → (∀ k → f k ≤ t)
    → Lim c f ≤ t
```

This relation is reflexive:

```
≤-refl : ∀ t → t ≤ t
≤-refl Z = ≤-Z
≤-refl (↑ t) = ≤-sucMono (≤-refl t)
≤-refl (Lim c f)
  = ≤-limiting f (λ k → ≤-cocone f k (≤-refl (f k)))
```

Crucially, it is also transitive, making the relation a pre-order. We modify our the order relation from that of Kraus et al. [2023] so that transitivity can be proven constructively, rather than adding it as a constructor for the relation. This allows us to prove well-foundedness of the relation without needing quotient types or other advanced features.

```
≤-trans : ∀ {t1 t2 t3} → t1 ≤ t2 → t2 ≤ t3 → t1 ≤ t3
≤-trans ≤-Z p23 = ≤-Z
≤-trans (≤-sucMono p12) (≤-sucMono p23)
  = ≤-sucMono (≤-trans p12 p23)
≤-trans p12 (≤-cocone f k p23)
  = ≤-cocone f k (≤-trans p12 p23)
≤-trans (≤-limiting f x) p23
```

<sup>1</sup>Not to be confused with Higher Inductive Types (HITs) from Homotopy Type Theory [Univalent Foundations Program 2013]

<sup>2</sup>Technically, this is a well-founded quasi-ordering because there are pairs of trees which are related by both  $\leq$  and  $\geq$ , but which are not propositionally equal.

```
= ≤-limiting f (λ k → ≤-trans (x k) p23)
≤-trans (≤-cocone f k p12) (≤-limiting .f x)
= ≤-trans p12 (x k)
```

We create an infix version of transitivity for more readable construction of proofs:

```
_≤;_ : ∀ {t1 t2 t3} → t1 ≤ t2 → t2 ≤ t3 → t1 ≤ t3
lt1 ≤; lt2 = ≤-trans lt1 lt2
```

**2.1.1 Strict Ordering.** We can define a strictly-less-than relation in terms of our less-than relation and the successor constructor:

```
_<_ : Tree → Tree → Set ℓ
t1 < t2 = ↑ t1 ≤ t2
```

That is, a  $t_1$  is strictly smaller than  $t_2$  if the tree one-size larger than  $t_1$  is as small as  $t_2$ . This relation has the properties one expects of a strictly-less-than relation: it is a transitive sub-relation of the less-than relation, every tree is strictly less than its successor, and no tree is strictly smaller than zero. JE ▶TODO more?◀

```
≤↑t : ∀ t → t ≤ ↑ t
≤↑t Z = ≤-Z
≤↑t (↑ t) = ≤-sucMono (≤↑t t)
≤↑t (Lim c f)
  = ≤-limiting f λ k →
    (≤↑t (f k))
  ≤; (≤-sucMono (≤-cocone f k (≤-refl (f k))))
```

```
<-in-≤ : ∀ {x y} → x < y → x ≤ y
<-in-≤ pf = (≤↑t _) ≤; pf
<-≤-in-≤ : ∀ {x y z} → x < y → y ≤ z → x < z
<-≤-in-≤ x< y≤z = x< y≤; y≤z
≤◦<-in-≤ : ∀ {x y z} → x ≤ y → y < z → x < z
≤◦<-in-≤ {x} {y} {z} x≤y y<z = (≤-sucMono x≤y) ≤; y<z
¬<Z : ∀ t → ¬(t < Z)
¬<Z t ()
```

## 2.2 Well Founded Induction

Recall the definition of a constructive well founded relation:

```
data Acc {A : Set a} ( _<_ : A → A → Set ℓ ) (x : A) : Set (a ⊔ ℓ) where
  acc : (rs : ∀ y → y < x → Acc _<_ y) → Acc _<_ x
WellFounded : (A → A → Set ℓ) → Set _
WellFounded _<_ = ∀ x → Acc _<_ x
```

That is, an element of a type is accessible for a relation if all strictly smaller elements of it are also accessible. A relation is well founded if all values are accessible with respect to that relation. This can then be used to define induction with arbitrary recursive calls on smaller values:

```

331 wfRec : (P : A → Set ℓ)
332   → (∀ x → ((y : A) → y < x → P y) → P x)
333   → ∀ x → P x

```

Following the construction of Kraus et al. [2023], we can show that the strict ordering on Brouwer trees is well founded. First, we prove a helper lemma: if a value is accessible, then all (not necessarily strictly) smaller terms are also accessible.

```

340 smaller-accessible : (x : Tree)
341   → Acc _<_ x → ∀ y → y ≤ x → Acc _<_ y
342 smaller-accessible x (acc r) y x<y
343   = acc (λ y' y'<y → r y' (<≤-in-< y'<y x<y))

```

Then we use structural reduction to show that all terms are accessible. The key observations are that zero is trivially accessible, since no trees are strictly smaller than it, and that the only way to derive  $\uparrow t \leq (\text{Lim } c f)$  is with `≤-cocone`, yielding a concrete index  $k$  for which  $\uparrow t \leq f k$ , on which we can recur.

```

352 ordWF : WellFounded _<_
353 ordWF Z = acc λ _ ()
354 ordWF (↑ x)
355   = acc (λ { y (≤-sucMono y≤x)
356     → smaller-accessible x (ordWF x) y y≤x})
357 ordWF (Lim c f) = acc helper
358   where
359     helper : (y : Tree) → (y < Lim c f)
360     → Acc _<_ y
361     helper y (≤-cocone .f k y<fk)
362       = smaller-accessible (f k)
363       (ordWF (f k)) y (<≤-in-≤ y<fk)

```

### 3 First Attempts at a Join

In this section, we present two faulty implementations of a join operator for trees. The first uses limits to define the join, but does not satisfy strict monotonicity. The second is defined inductively. Its satisfies strict monotonicity, but fails to be the least of all upper bounds, and requires us to assume that limits are only taken over non-empty types. In ??, we define SMB-trees a refinement of Brouwer trees that combines the benefits of both versions of the maximum.

#### 3.1 Limit-based Maximum

Since the limit constructor finds the least upper bound of the image of a function, it should be possible to define the maximum of two trees as a special case of general limits. Indeed, we can compute the maximum of  $t_1$  and  $t_2$  as the limit of the function that produces  $t_1$  when given 0 and  $t_2$  otherwise.

```

386 limMax : Tree → Tree → Tree
387 limMax t1 t2 = INLim λ n → if0 n t1 t2

```

This version of the maximum has several of the properties we want from a maximum function: it is monotone, idempotent, commutative, and is a true least-upper-bound of its inputs.

```

393 limMax≤L : ∀ {t1 t2} → t1 ≤ limMax t1 t2
394 limMax≤L {t1} {t2}
395   = ≤-cocone _ (Iso.inv CNIso 0)
396   (subst
397     (λ x → t1 ≤ if0 x t1 t2)
398     (sym (Iso.rightInv CNIso 0))
399     (≤-refl t1))

```

```

401 limMax≤R : ∀ {t1 t2} → t2 ≤ limMax t1 t2
402 -- Symmetric

```

```

403 limMaxIdem : ∀ {t} → limMax t t ≤ t
404 limMaxIdem {t} = ≤-limiting _ helper
405   where
406     helper : ∀ k → if0 (Iso.fun CNIso k) t t ≤ t
407     helper k with Iso.fun CNIso k
408     ... | zero = ≤-refl t
409     ... | suc n = ≤-refl t

```

**JE** ▶**TODO update description**◀ From these properties, we can compute several other useful properties: monotonicity, commutativity, and that it is in fact the least of all upper bounds.

```

416 limMaxMono : ∀ {t1 t2 t'1 t'2}
417   → t1 ≤ t'1 → t2 ≤ t'2
418   → limMax t1 t2 ≤ limMax t'1 t'2

```

```

420 limMaxCommut : ∀ {t1 t2} → limMax t1 t2 ≤ limMax t2 t1

```

```

421 limMaxLUB : ∀ {t1 t2 t} → t1 ≤ t → t2 ≤ t → limMax t1 t2 ≤ t

```

It is not surprising that this version of the maximum is a least upper bound: by definition `Lim` computes the least upper bound of a function's image, and `limMax` is simply `Lim` applied to a function whose image has (at most) two elements.

**3.1.1 Limitation: Strict Monotonicity.** The one crucial property that this formulation lacks is that it is not strictly monotone: we cannot deduce  $\text{max } t_1 t_1 < \text{max } t'_1 t'_2$  from  $t_1 < t'_1$  and  $t_2 < t'_2$ . This is because the only way to construct a proof that  $\uparrow t \leq \text{Lim } c f$  is using the `≤-cocone` constructor. So we would need to prove that  $\uparrow(\text{max } t_1 t_2) \leq t'_1$  or that  $\uparrow(\text{max } t_1 t_2) \leq t'_2$ , which cannot be deduced from the premises alone. What we want is to have  $\uparrow \text{max } (t_1) t_2 \leq \text{max } (\uparrow t_1) (\uparrow t_2)$ , so that strict monotonicity is a direct consequence of ordinary monotonicity of the maximum. This is not possible when defining the constructor as a limit.



### 3.2 Recursive Maximum

In our next attempt at defining a maximum operator, we obtain strict monotonicity by making  $\text{indMax } (\uparrow t_1) (\uparrow t_2) = \uparrow(\text{indMax } t_1 t_2)$  hold definitionally. Then, provided  $\text{indMax}$  is monotone, it will also be strictly monotone.

To do this, we compute the maximum of two trees recursively, pattern matching on the operands. We use a *view* [?] datatype to identify the cases we are matching on: we are matching on two arguments, which each have three possible constructors, but several cases overlap. Using a view type lets us avoid enumerating all nine possibilities when defining the maximum and proving its properties.

To begin, we parameterize our definition over a function yielding some element for any code's type.

```

module IndMax {ℓ}
  (C : Set ℓ)
  (El : C → Set ℓ)
  (CN : C) (CNIso : Iso (El CN) ℕ)
  (default : (c : C) → El c) where

  We then define our view type:

private
  data IndMaxView : Tree → Tree → Set ℓ where
    IndMaxZ-L : ∀ {t} → IndMaxView Z t
    IndMaxZ-R : ∀ {t} → IndMaxView t Z
    IndMaxLim-L : ∀ {t} {c : C} {f : El c → Tree}
      → IndMaxView (Lim c f) t
    IndMaxLim-R : ∀ {t} {c : C} {f : El c → Tree}
      → (∀ {c' : C} {f' : El c' → Tree} → ¬(t = Lim c' f'))
      → IndMaxView t (Lim c f)
    IndMaxLim-Suc : ∀ {t1 t2} → IndMaxView (↑ t1) (↑ t2)

```

opaque

$\text{indMaxView} : \forall t_1 t_2 \rightarrow \text{IndMaxView } t_1 t_2$

Our view type has five cases. The first two handle when either input is zero, and the second two handle when either input is a limit. The final case is when both inputs are successors. *indMaxView* computes the view for any pair of trees.

The maximum is then defined by pattern matching on the view for its arguments:

```

indMax : Tree → Tree → Tree
indMax' : ∀ {t1 t2} → IndMaxView t1 t2 → Tree

indMax t1 t2 = indMax' (indMaxView t1 t2)
indMax' {Z} {t2} IndMaxZ-L = t2
indMax' {t1} {Z} IndMaxZ-R = t1
indMax' {(Lim c f)} {t2} IndMaxLim-L
  = Lim c λ x → indMax (f x) t2
indMax' {t1} {(Lim c f)} (IndMaxLim-R _)
  = Lim c (λ x → indMax t1 (f x))
indMax' {(↑ t1)} {(↑ t2)} IndMaxLim-Suc = ↑ (indMax t1 t2)

```

The maximum of zero and  $t$  is always  $t$ , and the maximum of  $t$  and the limit of  $f$  is the limit of the function computing the maximum between  $t$  and  $f x$ . Finally, the maximum of two successors is the successor of the two maxima, giving the definitional equality we need for strict monotonicity.

This definition only works when limits of all codes are inhabited. The  $\leq$ -limiting constructor means that  $\text{Lim } c f \leq Z$  whenever  $\text{El } c$  is uninhabited. So  $\text{indMax } \uparrow Z \text{ Lim } c f$  will not actually be an upper bound for  $\uparrow Z$  if  $c$  has no inhabitants. In ?? we show how to circumvent this restriction.

Under the assumption that all code are inhabited, we obtain several of our desired properties for a maximum: it is an upper bound, it is monotone and strictly monotonicity, and it is associative and commutative.

opaque

unfolding indMax indMax'

```

indMax-≤L : ∀ {t1 t2} → t1 ≤ indMax t1 t2
indMax-≤L {t1} {t2} with indMaxView t1 t2
... | IndMaxZ-L = ≤-Z
... | IndMaxZ-R = ≤-refl _
... | IndMaxLim-L {f = f}
  = extLim f (λ x → indMax (f x) t2) (λ k → indMax-≤L)
... | IndMaxLim-R {f = f} _
  = underLim λ k → indMax-≤L {t2 = f k}
... | IndMaxLim-Suc
  = ≤-sucMono indMax-≤L

```

$\text{indMax-≤R} : \forall \{t_1 t_2\} \rightarrow t_2 \leq \text{indMax } t_1 t_2$

-- Symmetric

$\text{indMax-monoL} : \forall \{t_1 t'_1 t_2\} \rightarrow t_1 \leq t'_1 \rightarrow \text{indMax } t_1 t_2 \leq \text{indMax } t'_1 t_2$

$\text{indMax-monoR} : \forall \{t_1 t_2 t'_2\} \rightarrow t_2 \leq t'_2 \rightarrow \text{indMax } t_1 t_2 \leq \text{indMax } t_1 t'_2$

$\text{indMax-mono} : \forall \{t_1 t_2 t'_1 t'_2\} \rightarrow t_1 \leq t'_1 \rightarrow t_2 \leq t'_2 \rightarrow \text{indMax } t_1 t_2 \leq \text{indMax } t'_1 t'_2$

$\text{indMax-strictMono} : \forall \{t_1 t_2 t'_1 t'_2\} \rightarrow t_1 < t'_1 \rightarrow t_2 < t'_2 \rightarrow \text{indMax } t_1 t_2 < \text{indMax } t'_1 t'_2$

$\text{indMax-strictMono } lt1 lt2 = \text{indMax-mono } lt1 lt2$

$\text{indMax-assocL} : \forall t_1 t_2 t_3 \rightarrow \text{indMax } t_1 (\text{indMax } t_2 t_3) \leq \text{indMax } (\text{indMax } t_1 t_2) t_3$

$\text{indMax-assocR} : \forall t_1 t_2 t_3 \rightarrow \text{indMax } (\text{indMax } t_1 t_2) t_3 \leq \text{indMax } t_1 (\text{indMax } t_2 t_3)$

$\text{indMax-commut} : \forall t_1 t_2 \rightarrow \text{indMax } t_1 t_2 \leq \text{indMax } t_2 t_1$

**3.2.1 Limitation: Idempotence.** The problem with an inductive definition of the maximum is that we cannot prove that it is idempotent. Since `indMax` is associative and commutative, proving idempotence is equivalent to proving that it computes a true least-upper-bound.

The difficulty lies in showing that `indMax (Lim c f) (Lim c f) ≤ (Lim c f)`. By our definition, `indMax (Lim c f) (Lim c f)` reduces to

$$(\text{Lim } c \lambda x \rightarrow (\text{Lim } c \lambda y \rightarrow \text{indMax } (f \ x) \ (f \ y))) \leq \text{Lim } c \ f$$

We cannot use `≤-cocone` to prove this, since the left hand side is not necessarily equal to `f k` for any `k : El c`. So the only possibility is to use `≤-limiting`. Applying it twice, along with a use of commutativity of `indMax`, we are left with the following goal:

$$(\forall x \rightarrow (\forall y \rightarrow \text{indMax } (f \ x) \ (f \ y))) \leq \text{Lim } c \ f$$

There is no a priori way to prove this goal without already having a proof that `indMax` is a least upper bound. But proving that was the whole point of proving idempotence! An inductive hypothesis would give that `indMax (f x) (f x) ≤ f x ≤ Lim c f`, but it does not apply when the arguments to `indMax` are not equal. Because we are working with constructive ordinals, we have no trichotomy property [?], and hence no guarantee that `indMax (f x) (f y)` will be one of `f x` and `f y`.

We now have two competing definitions for the maximum: the limit version, which is not strictly monotone, and the inductive version, which is not actually a least upper bound. In the next section, we describe a large class of trees for which `indMax` is idempotent, and hence does compute a true upper bound. We then use that in ?? to create a version of ordinals whose join has the best properties of both `limMax` and `indMax`. JE ▶ TODO recall the algebraic definition of semilattice◀

## 4 Trees with a Strictly-Monotone Idempotent Join

### 4.1 Well-Behaved Trees

Our first step in defining an ordinal notation with a well behaved maximum is to identify a class of Brouwer trees which are well behaved with respect to the inductive maximum. As we saw in

The answer, it turns out, is more limits: if we `indMax` a term with itself an infinite number of times, the result will be idempotent with respect to `indMax`. First, we define a function to `indMax` a term with itself `n` times or a given number `n`:

$$\text{nindMax} : \text{Tree} \rightarrow \mathbb{N} \rightarrow \text{Tree}$$

$$\text{nindMax } t \ \mathbb{N}.\text{zero} = Z$$

$$\text{nindMax } t \ (\mathbb{N}.\text{suc } n) = \text{indMax } (\text{nindMax } t \ n) \ t$$

To compute a tree equivalent to the infinite chain of applications `indMax t (indMax t (indMax t ...))`, we take the limit of `n` applications over all `n`:

$$\text{indMax}\infty : \text{Tree} \rightarrow \text{Tree}$$

$$\text{indMax}\infty \ t = \mathbb{N}\text{Lim } (\lambda n \rightarrow \text{nindMax } t \ n)$$

This operator has useful basic properties: it is monotone, and it computes an upper bound on its argument.

$$\text{indMax}\infty\text{-self} : \forall t \rightarrow t \leq \text{indMax}\infty \ t$$

$$\text{indMax}\infty\text{-mono} : \forall \{t_1 \ t_2\}$$

$$\rightarrow t_1 \leq t_2$$

$$\rightarrow (\text{indMax}\infty \ t_1) \leq (\text{indMax}\infty \ t_2)$$

However, the most important property we want from `indMax` is that `indMax` is idempotent with respect to it. The first step to showing this is realizing that we can take the maximum of `t` and `indMax` `t` and we have a tree that is no larger than `indMax` `t`: because it is already an infinite chain of applications, adding one more makes no difference.

$$\text{indMax}\infty\text{-lt1} : \forall t \rightarrow \text{indMax } (\text{indMax}\infty \ t) \ t \leq \text{indMax}\infty \ t$$

$$\text{indMax}\infty\text{-lt1 } t = \leq\text{-limiting } \_ \lambda k \rightarrow \text{helper } (\text{Iso.fun } \text{CNIso } k)$$

where

$$\text{helper} : \forall n \rightarrow \text{indMax } (\text{nindMax } t \ n) \ t \leq \text{indMax}\infty \ t$$

$$\text{helper } n =$$

$$\leq\text{-cocone } \_ (\text{Iso.inv } \text{CNIso } (\mathbb{N}.\text{suc } n))$$

$$(\text{subst } (\lambda sn \rightarrow \text{nindMax } t \ (\mathbb{N}.\text{suc } n) \leq \text{nindMax } t \ sn))$$

$$(\text{sym } (\text{Iso.rightInv } \text{CNIso } (\text{suc } n)))$$

$$(\leq\text{-refl } \_)$$

If adding one more `indMax t` has no effect, then adding `n` more will also have no effect:

$$\text{indMax}\infty\text{-ltn} : \forall n \ t$$

$$\rightarrow \text{indMax } (\text{indMax}\infty \ t) \ (\text{nindMax } t \ n) \leq \text{indMax}\infty \ t$$

$$\text{indMax}\infty\text{-ltn } \mathbb{N}.\text{zero } t = \text{indMax}\leq Z \ (\text{indMax}\infty \ t)$$

$$\text{indMax}\infty\text{-ltn } (\mathbb{N}.\text{suc } n) \ t =$$

$$\text{indMax-monoR } (\text{indMax-commut } (\text{nindMax } t \ n) \ t)$$

$$\leq \text{indMax-assocL } (\text{indMax}\infty \ t) \ t \ (\text{nindMax } t \ n)$$

$$\leq \text{indMax-monoL } (\text{indMax}\infty\text{-lt1 } t)$$

$$\leq \text{indMax}\infty\text{-ltn } n \ t$$

By our inductive definition of `indMax`, we have that

$$\text{indMax } (\text{indMax}\infty \ t) (\text{indMax}\infty \ t)$$

is equal to

$$\mathbb{N}\text{Lim } (\lambda n \rightarrow \text{indMax } (\text{nindMax } n \ t) \ (\text{indMax}\infty \ t))$$

Our previous lemma gives that, for any `n`, `indMax` `t` is an upper bound for `indMax (nindMax n t) (indMax` `t)`. So `≤-limiting` gives that the limit over all `n` is also bounded by `indMax` `t`, i.e. `Lim` constructs the least of all upper bounds. This gives us our key result: up to `≤`, `indMax` is idempotent on values constructed with `indMax`.

```

661 indMax $\infty$ -idem :  $\forall t$ 
662    $\rightarrow$  indMax (indMax $\infty$  t) (indMax $\infty$  t)  $\leq$  indMax $\infty$  t
663 indMax $\infty$ -idem t =
664    $\leq$ -limiting _  $\lambda k \rightarrow$ 
665     (indMax-commut (nindMax t (Iso.fun CNIso k)) (indMax $\infty$  t))
666    $\leq$  ; indMax- $\infty$ lt (Iso.fun CNIso k) t

```

There is one last property to prove that will be useful in the next section: `indMax $\infty$  t` is a lower bound on  $t$ , and hence equivalent to it, whenever `indMax` is idempotent on  $t$ . If taking `indMax` of  $t$  with itself does not increase its size, doing so  $n$  times will not increase its size, so again the result follows from `Lim` being the least upper bound.

```

674 indMax $\infty$ - $\leq$  :  $\forall \{t\} \rightarrow$  indMax t t  $\leq$  t  $\rightarrow$  indMax $\infty$  t  $\leq$  t
675 indMax $\infty$ - $\leq$  lt =  $\leq$ -limiting _  $\lambda k \rightarrow$  nindMax- $\leq$  (Iso.fun CNIso k) lt

```

```

676 where
677   nindMax- $\leq$  :  $\forall \{t\} n \rightarrow$  indMax t t  $\leq$  t  $\rightarrow$  nindMax t n  $\leq$  t
678   nindMax- $\leq$  N.zero lt =  $\leq$ -Z
679   nindMax- $\leq$  {t = t} (N.suc n) lt = (indMax-monoL {t1 = nindMax

```

An immediate corollary of this is that `indMax $\infty$  (indMax $\infty$  t)` is equivalent to `indMax $\infty$  t`.

## 4.2 Strictly Monotone Brouwer Trees

Now that we have identified a substantial class of well behaved Brouwer trees, we want to define a new type containing only those trees. These are SMB-trees: strictly monotone Brouwer trees. In this section, we will define them, and show how they can be given a similar interface to Brouwer trees.

To begin, we declare a new Agda module, with the same parameters we have been working with thus far: a type of codes, interpretations of those codes into types, and a code whose interpretation is isomorphic to  $\mathbb{N}$ .

```

694 module Idem { $\ell$ }
695   (C : Set  $\ell$ )
696   (El : C  $\rightarrow$  Set  $\ell$ )
697   (CN : C) (CNIso : Iso (El CN) N) where

```

We import all of our definitions so far, using the “Brouwer” prefix to distinguish them from the trees and ordering we are about to define. Critically, we do not instantiate these with the same interpretation function. Instead, we interpret each code wrapped in `Maybe`. This ensures that we always take Brouwer limits over non-empty sets, an assumption that was critical for the definitions of `??`. However, we place no such restriction on SMB-trees.

```

707 import Brouwer C ( $\lambda c \rightarrow$  Maybe (El c)) CN (maybeNatIso CNIso) as Brouwer

```

**4.2.1 Refining Brouwer Trees.** We define SMB-trees as a dependent record, containing an underlying Brouwer tree, and a proof that `indMax` is idempotent on this tree.

```

713 record SMBTree : Set  $\ell$  where
714   constructor MkTree

```

```

field
rawTree : Brouwer.Tree
isIdem : (indMax rawTree rawTree) Brouwer. $\leq$  rawTree
open SMBTree

```

We can then define so-called “smart-constructors” corresponding to each of the constructors for Brouwer-trees: zero, successor, and limit. Zero and successor directly correspond to the Brouwer tree zero and successor. Their proofs of idempotence are trivial from the properties of Brouwer  $\leq$ .

```

opaque
unfolding indMax

```

```

Z : SMBTree
Z = MkTree Brouwer.Z Brouwer. $\leq$ -Z

```

```

 $\uparrow$  : SMBTree  $\rightarrow$  SMBTree

```

```

 $\uparrow$  (MkTree t pf) = MkTree (Brouwer. $\uparrow$  t) (Brouwer. $\leq$ -sucMono pf)

```

However, constructing the limit of a sequence of SMB-trees is not so easy. Since we instantiated `El` to wrap its result in `Maybe`, we need to handle *nothing* for each limit, but we can use `Z` as a default value, since adding it to any sequence does not change the least upper bound. More challenging is how, as we saw in `??`, Brouwer trees do not have `indMax (Lim c f) (Lim c f)  $\leq$  Lim c f`, so we cannot directly produce a proof of idempotence.

Our key insight is to define limits of SMB-trees using `indMax $\infty$`  on the underlying trees: for any function producing SMB-trees, we take the limit of the underlying trees, then `indMax` that result with itself an infinite number of times. The idempotence proof is then the property of `indMax $\infty$`  that we proved in `??`.

```

Lim :  $\forall (c : C) \rightarrow (f : El c \rightarrow$  SMBTree  $\rightarrow$  SMBTree

```

```

Lim c f =

```

```

  MkTree

```

```

  (indMax $\infty$  (Brouwer.Lim c (maybe' ( $\lambda x \rightarrow$  rawTree (f x)) Brouwer.Z)))

```

```

  (indMax $\infty$ -idem _)

```

**4.2.2 Ordering SMB-trees.** SMB-trees are ordered by the order on their underlying Brouwer trees:

```

record  $\_ \leq \_$  (t1 t2 : SMBTree) : Set  $\ell$  where

```

```

  constructor mk $\leq$ 

```

```

  inductive

```

```

  field

```

```

  gets $\leq$  : (rawTree t1) Brouwer. $\leq$  (rawTree t2)

```

```

open  $\_ \leq \_$ 

```

Having a successor function allows us to define a strict ordering on SMB-trees.

```

 $\_ < \_$  : SMBTree  $\rightarrow$  SMBTree  $\rightarrow$  Set  $\ell$ 

```

```

 $\_ < \_$  t1 t2 = ( $\uparrow$  t1)  $\leq$  t2

```

The next step is to prove that our SMB-tree constructors satisfy the same inequalities as Brouwer trees. Since SMB-trees are ordered by their underlying Brouwer trees, most properties can be directly lifted from Brouwer trees to SMB-trees.

opaque

unfolding  $Z \uparrow$

$\leq \uparrow : \forall t \rightarrow t \leq \uparrow t$

$\leq \uparrow t = \text{mk} \leq (\text{Brouwer}.\leq \uparrow t \_)$

$\leq \_ \_ : \forall \{t_1 \ t_2 \ t_3\} \rightarrow t_1 \leq t_2 \rightarrow t_2 \leq t_3 \rightarrow t_1 \leq t_3$

$\leq \_ \_ (\text{mk} \leq lt1) (\text{mk} \leq lt2) = \text{mk} \leq (\text{Brouwer}.\leq\text{-trans } lt1 \ lt2)$

$\leq\text{-refl} : \forall \{t\} \rightarrow t \leq t$

$\leq\text{-refl} = \text{mk} \leq (\text{Brouwer}.\leq\text{-refl } \_)$

The constructors for  $\leq$  each have a counterpart for SMB-trees. For zero and successor, these are trivially lifted.

$\leq\text{-Z} : \forall \{t\} \rightarrow Z \leq t$

$\leq\text{-Z} = \text{mk} \leq \text{Brouwer}.\leq\text{-Z}$

$\leq\text{-sucMono} : \forall \{t_1 \ t_2\} \rightarrow t_1 \leq t_2 \rightarrow \uparrow t_1 \leq \uparrow t_2$

$\leq\text{-sucMono} (\text{mk} \leq lt) = \text{mk} \leq (\text{Brouwer}.\leq\text{-sucMono } lt)$

The constructors for ordering limits require more attention. To show that an SMB-tree limit is an upper-bound, we use the fact that the underlying limit was an upper bound, and the fact that  $\text{indMax}^\infty$  is an upper bound, since the SMB-tree  $\text{Lim}$  wraps its result in  $\text{indMax}^\infty$ . Note that, since we already have transitivity for our new  $\leq$ , we can simply show that  $f \ k$  is less than the limit of  $f$ , avoiding the more complicated form of  $\leq\text{-cocone}$ .

$\leq\text{-limUpperBound} : \forall \{c : \mathcal{C}\} \rightarrow \{f : \text{El } c \rightarrow \text{SMBTree}\}$

$\rightarrow \forall k \rightarrow f \ k \leq \text{Lim } c \ f$

$\leq\text{-limUpperBound } \{c = c\} \{f = f\} \ k$

$= \text{mk} \leq (\text{Brouwer}.\leq\text{-cocone } \_ (\text{just } k) (\text{Brouwer}.\leq\text{-refl } \_))$

$\text{Brouwer}.\leq \ ; \text{indMax}^\infty\text{-self } (\text{Brouwer}.\text{Lim } c \ \_))$

Finally, we need to show that the SMT-tree limit is less than all other upper bounds. Suppose  $t : \text{SMBTree}$  is an upper bound for  $f$ , and  $t_u$  is the underlying tree for  $t$ , and  $f_u$  computes the underlying trees for  $f$ . Then  $\leq\text{-limiting}$  gives that the underlying tree for  $t$  is an upper bound for the trees underlying the image of  $f$ . However, the SMB-tree limit wraps its result in  $\text{indMax}^\infty$ . The monotonicity of  $\text{indMax}^\infty$  then gives that  $\text{indMax}(\text{Lim } c \ f_u)$  is less than  $\text{indMax}^\infty \ t'$ . In ??, we showed that  $\text{indMax}^\infty$  had no effect on Brouwer trees that  $\text{indMax}$  was idempotent on. This is exactly what the  $\text{isIdem}$  field of SMB-trees contains! So we have  $\text{indMax}^\infty \ t' \leq t'$ , and transitivity gives our result.

$\leq\text{-limLeast} : \forall \{c : \mathcal{C}\} \rightarrow \{f : \text{El } c \rightarrow \text{SMBTree}\}$

$\rightarrow \{t : \text{SMBTree}\}$

$\rightarrow (\forall k \rightarrow f \ k \leq t) \rightarrow \text{Lim } c \ f \leq t$

$\leq\text{-limLeast } \{f = f\} \{t = \text{MkTree } t \text{ idem}\} \ lt$

$= \text{mk} \leq (\$

$\text{indMax}^\infty\text{-mono}$

$(\text{Brouwer}.\leq\text{-limiting } \_)$

$(\text{maybe } (\lambda k \rightarrow \text{get} \leq (lt \ k)) \ \text{Brouwer}.\leq\text{-Z}))$

$\text{Brouwer}.\leq \ ; (\text{indMax}^\infty\text{-self } \text{idem}) \ )$

**4.2.3 The Join for SMB-trees.** Our whole reason for defining SMB-trees was to define a well-behaved maximum operator, and we finally have the tools to do so. We can define the join in terms of  $\text{indMax}$  on the underlying trees. The proof that the  $\text{indMax}$  is idempotent on the result follows from associativity, commutativity, and monotonicity of  $\text{indMax}$ .

opaque

unfolding  $\text{indMax } Z \uparrow \text{indMaxView}$

$\text{max} : \text{SMBTree} \rightarrow \text{SMBTree} \rightarrow \text{SMBTree}$

$\text{max } t_1 \ t_2 =$

$\text{MkTree}$

$(\text{indMax } (\text{rawTree } t_1) (\text{rawTree } t_2))$

$(\text{indMax}\text{-swap4}$

$\text{Brouwer}.\leq \ ; \text{indMax}\text{-mono } (\text{isIdem } t_1) (\text{isIdem } t_2))$

For Brouwer trees,  $\text{indMax}$  had all the properties we wanted except for idempotence. All of these can be lifted directly to SMB-trees:

$\text{max}\text{-}\leq L : \forall \{t_1 \ t_2\} \rightarrow t_1 \leq \text{max } t_1 \ t_2$

$\text{max}\text{-}\leq R : \forall \{t_1 \ t_2\} \rightarrow t_2 \leq \text{max } t_1 \ t_2$

$\text{max}\text{-mono} : \forall \{t_1 \ t'_1 \ t_2 \ t'_2\} \rightarrow t_1 \leq t'_1 \rightarrow t_2 \leq t'_2 \rightarrow$

$\text{max } t_1 \ t_2 \leq \text{max } t'_1 \ t'_2$

$\text{max}\text{-idem} \leq : \forall \{t\} \rightarrow t \leq \text{max } t \ t$

$\text{max}\text{-commut} : \forall t_1 \ t_2 \rightarrow \text{max } t_1 \ t_2 \leq \text{max } t_2 \ t_1$

$\text{max}\text{-assocL} : \forall t_1 \ t_2 \ t_3 \rightarrow \text{max } t_1 (\text{max } t_2 \ t_3) \leq \text{max } (\text{max } t_1 \ t_2) \ t_3$

$\text{max}\text{-assocR} : \forall t_1 \ t_2 \ t_3 \rightarrow \text{max } (\text{max } t_1 \ t_2) \ t_3 \leq \text{max } t_1 (\text{max } t_2 \ t_3)$

In particular,  $\text{max}$  is strictly monotone, and distributes over the successor:

$\text{max}\text{-strictMono} : \forall \{t_1 \ t'_1 \ t_2 \ t'_2 : \text{SMBTree}\}$

$\rightarrow t_1 < t'_1 \rightarrow t_2 < t'_2 \rightarrow \text{max } t_1 \ t_2 < \text{max } t'_1 \ t'_2$

$\text{max}\text{-sucMono} : \forall \{t_1 \ t_2 \ t'_1 \ t'_2 : \text{SMBTree}\}$

$\rightarrow \text{max } t_1 \ t_2 \leq \text{max } t'_1 \ t'_2 \rightarrow \text{max } t_1 \ t_2 < \text{max } (\uparrow t'_1) (\uparrow t'_2)$

However, because we restricted SMB-trees to only contain Brouwer trees that  $\text{indMax}$  is idempotent on, we can prove that  $\text{Max}$  is idempotent for SMB-trees:

$\text{max}\text{-idem} : \forall \{t : \text{SMBTree}\} \rightarrow \text{max } t \ t \leq t$

$\text{max}\text{-idem } \{t = \text{MkTree } t \text{ pf}\} = \text{mk} \leq \text{pf}$

These together are enough to prove that our maximum is the least of all upper bounds.

$\text{max}\text{-LUB} : \forall \{t_1 \ t_2 \ t\} \rightarrow t_1 \leq t \rightarrow t_2 \leq t \rightarrow \text{max } t_1 \ t_2 \leq t$

$\text{max}\text{-LUB } lt1 \ lt2 = \text{max}\text{-mono } lt1 \ lt2 \leq \ ; \text{max}\text{-idem}$



```

881 ≤-extLim : ∀ {c : C} → {f1 f2 : El c → SMBTree}
882   → (∀ k → f1 k ≤ f2 k)
883   → Lim c f1 ≤ Lim c f2
884 ≤-extLim lt = ≤-limLeast (λ k → lt k ≤ ; ≤-limUpperBound k)
885
886 ≤-extExists : ∀ {c1 c2 : C} → {f1 : El c1 → SMBTree} {f2 : El c2 → SMBTree}
887   → (∀ k1 → Σ [ k2 ∈ El c2 ] f1 k1 ≤ f2 k2)
888   → Lim c1 f1 ≤ Lim c2 f2
889 ≤-extExists {f1 = f1} {f2} lt = ≤-limLeast (λ k1 → proj2 (lt k1) ≤ ; ≤-limUpperBound (proj1 (lt k1)))
890
891 ¬Z<↑ : ∀ t → ¬ ((↑ t) ≤ Z)
892 ¬Z<↑ t pf = Brouwer.¬<Z (rawTree t) (get≤ pf)
893
894 max-≤L = mk≤ indMax-≤L
895 max-≤R = mk≤ indMax-≤R
896
897 max-mono lt1 lt2 = mk≤ (indMax-mono (get≤ lt1) (get≤ lt2))
898
899 max-monoR : ∀ {t1 t2 t'2} → t2 ≤ t'2 → max t1 t2 ≤ max t1 t'2
900 max-monoR {t1} {t2} {t'2} lt = max-mono {t1 = t1} {t'1 = t1} {t2 = t2} {t'2 = t'2} (≤-refl {t1}) lt
901
902 max-monoL : ∀ {t1 t'1 t2} → t1 ≤ t'1 → max t1 t2 ≤ max t'1 t2
903 max-monoL {t1} {t'1} {t2} lt = max-mono {t1} {t'1} {t2} {t2} lt (≤-refl {t2})
904
905 max-idem≤ {t = MkTree t pf} = max-≤L
906
907 max-commut t1 t2 = mk≤ (indMax-commut (rawTree t1) (rawTree t2))
908
909 max-assocL t1 t2 t3 = mk≤ (indMax-assocL _ _ _)
910
911 max-assocR t1 t2 t3 = mk≤ (indMax-assocR _ _ _)
912
913 max-swap4 : ∀ {t1 t'1 t2 t'2} → max (max t1 t'1) (max t2 t'2) ≤ max (max t2 t'2) (max t1 t'1)
914 max-swap4 = mk≤ indMax-swap4
915
916 max-strictMono lt1 lt2 = mk≤ (indMax-strictMono (get≤ lt1) (get≤ lt2))
917
918 max-sucMono lt = mk≤ (indMax-sucMono (get≤ lt))
919
920 INLim : (IN → SMBTree) → SMBTree
921 INLim f = Lim CN (λ cn → f (Iso.fun CNIso cn))
922
923 max' : SMBTree → SMBTree → SMBTree
924 max' t1 t2 = INLim (λ n → if0 n t1 t2)
925
926 max'-≤L : ∀ {t1 t2} → t1 ≤ max' t1 t2
927 max'-≤L {t1} {t2}
928   = subst (λ x → t1 ≤ if0 x t1 t2) (sym (Iso.rightInv CNIso 0))
929   ≤-limUpperBound (Iso.inv CNIso 0)
930
931 max'-≤R : ∀ {t1 t2} → t2 ≤ max' t1 t2
932 max'-≤R {t1} {t2}
933   = subst (λ x → t2 ≤ if0 x t1 t2) (sym (Iso.rightInv CNIso 1))
934   ≤-refl ≤
935   ≤-limUpperBound (Iso.inv CNIso 1)
936
937 max'-Idem : ∀ {t} → max' t t ≤ t
938 max'-Idem {t} = ≤-limLeast helper
939
940 where
941   helper : (k : El CN) →
942     Lim c (λ x → if0 (Iso.fun CNIso k) (f x) (g x)) ≤
943     if0 (Iso.fun CNIso k) (Lim c f) (Lim c g)
944   helper kn with Iso.fun CNIso kn
945   ... | zero = ≤-refl
946   ... | suc n = ≤-refl
947
948 max'-LUB : ∀ {t1 t2 t} → t1 ≤ t → t2 ≤ t → max' t1 t2 ≤ t
949 max'-LUB lt1 lt2 = max'-Mono lt1 lt2 ≤ ; max'-Idem
950
951 max≤max' : ∀ {t1 t2} → max t1 t2 ≤ max' t1 t2
952 max≤max' = max'-LUB max'-≤L max'-≤R
953
954 limSwap : ∀ {c1 c2} {f : El c1 → El c2 → SMBTree} → (Lim c1 λ x → Lim c2 λ y → f x y) → Lim c2 λ y → Lim c1 λ x → f x y
955 limSwap = ≤-limLeast (λ x → ≤-limLeast λ y → ≤-limUpperBound x ≤ ; ≤-limUpperBound y)
956
957 max-swapL : ∀ {c} {f g : El c → SMBTree} → Lim c (λ k → max (f k) (g k))
958 max-swapL {c} {f} {g} = ≤-extLim (λ k → max≤max') ≤ ; limSwap ≤ ; ≤-refl
959
960 where
961   helper : (k : El CN) →
962     Lim c (λ x → if0 (Iso.fun CNIso k) (f x) (g x)) ≤
963     if0 (Iso.fun CNIso k) (Lim c f) (Lim c g)
964   helper kn with Iso.fun CNIso kn
965   ... | zero = ≤-refl
966   ... | suc n = ≤-refl
967
968 max-swapR : ∀ {c} {f g : El c → SMBTree} → max (Lim c f) (Lim c g) ≤ Lim c (λ z → if0 (Iso.fun CNIso k) (f z) (g z))
969 max-swapR {c} {f} {g} = max≤max' ≤ ; ≤-extLim helper ≤ ; limSwap ≤ ; ≤-refl
970
971 where
972   helper : (k : El CN) →
973     Lim c (λ x → if0 (Iso.fun CNIso k) (f x) (g x)) ≤
974     if0 (Iso.fun CNIso k) (Lim c f) (Lim c g)
975   helper kn with Iso.fun CNIso kn
976   ... | zero = ≤-refl
977   ... | suc n = ≤-refl
978
979 max-swapL : ∀ {c} {f g : El c → SMBTree} → Lim c (λ k → max (f k) (g k))
980 max-swapL {c} {f} {g} = ≤-extLim (λ k → max≤max') ≤ ; limSwap ≤ ; ≤-refl
981
982 where
983   helper : (k : El CN) →
984     Lim c (λ x → if0 (Iso.fun CNIso k) (f x) (g x)) ≤
985     if0 (Iso.fun CNIso k) (Lim c f) (Lim c g)
986   helper kn with Iso.fun CNIso kn
987   ... | zero = ≤-refl
988   ... | suc n = ≤-refl
989
990 max-swapR : ∀ {c} {f g : El c → SMBTree} → max (Lim c f) (Lim c g) ≤ Lim c (λ z → if0 (Iso.fun CNIso k) (f z) (g z))
991 max-swapR {c} {f} {g} = max≤max' ≤ ; ≤-extLim helper ≤ ; limSwap ≤ ; ≤-refl
992
993 where
994   helper : (k : El CN) →
995     Lim c (λ x → if0 (Iso.fun CNIso k) (f x) (g x)) ≤
996     if0 (Iso.fun CNIso k) (Lim c f) (Lim c g)
997   helper kn with Iso.fun CNIso kn
998   ... | zero = ≤-refl
999   ... | suc n = ≤-refl

```

## References

- Guillaume Allais, Edwin Brady, Nathan Corbyn, Ohad Kammar, and Jeremy Yallop. 2023. Frex: dependently-typed algebraic simplification. arXiv:2306.15375 [cs.PL]
- Yves Bertot and Pierre Castéran. 2004. Interactive Theorem Proving and Program Development. Springer-Verlag.
- Marc Bezem and Thierry Coquand. 2022. Loop-checking and the uniform word problem for join-semilattices with an inflationary endomorphism. Theoretical Computer Science 913 (2022), 1–7. <https://doi.org/10.1016/j.tcs.2022.01.017>
- Jonathan H.W. Chan. 2022. Sized dependent types via extensional type theory. Master's thesis. University of British Columbia. <https://doi.org/10.14288/1.0416401>
- Nathan Corbyn. 2021. Proof Synthesis with Free Extensions in Intensional Type Theory. Technical Report. University of Cambridge. MEng Dissertation.
- Nicolai Kraus, Fredrik Nordvall Forsberg, and Chuangjie Xu. 2023. Type-theoretic approaches to ordinals. Theoretical Computer Science 957 (2023), 113843. <https://doi.org/10.1016/j.tcs.2023.113843>
- The Univalent Foundations Program. 2013. Homotopy Type Theory: Univalent Foundations of Mathematics. <https://homotopytypetheory.org/book>, Institute for Advanced Study.